
Contrôle S4 – Corrigé
Architecture des ordinateurs

Durée : 1 h 30

Répondre exclusivement sur le document réponse.

Exercice     1  (4 points)
Remplir le tableau présent sur le document réponse. Donnez le nouveau contenu des registres (sauf le PC)

et/ou de la mémoire modifiés par les instructions.  Vous utiliserez la représentation hexadécimale.  La

mémoire et les registres sont réinitialisés à chaque nouvelle instruction.

Valeurs initiales : D0 = $FFFF0015  A0 = $00005000  PC = $00006000

D1 = $12340004  A1 = $00005008

D2 = $FFFFFFE1  A2 = $00005010

$005000  54 AF 18 B9 E7 21 48 C0

$005008  C9 10 11 C8 D4 36 1F 88

$005010  13 79 01 80 42 1A 2D 49

Exercice     2  (3 points)
Remplir le tableau présent sur le  document réponse. Vous devez trouver le nombre manquant (sous sa

forme hexadécimale) en fonction de la taille de l’opération et de la valeur des flags après l’opération. Si

plusieurs solutions sont possibles, vous retiendrez uniquement la plus petite.

Exercice     3  (4 points)
Soit le programme ci-dessous. Complétez le tableau présent sur le document réponse.

Main        move.l  #$85A51000,d7

next1       moveq.l #1,d1
            cmpi.w  #$80,d7
            blt     next2
            moveq.l #2,d1

next2       move.l  d7,d2
            ror.l   #4,d2
            swap    d2
            rol.w   #8,d2
            rol.b   #4,d2

next3       clr.l   d3
            move.l  d7,d0
loop3       addq.l  #1,d3
            subq.w  #2,d0
            bne     loop3

next4       clr.l   d4
            move.l  d7,d0
loop4       addq.l  #1,d4
            dbra    d0,loop4      ; DBRA = DBF

Exercice     4  (9 points)
Toutes les questions de cet exercice sont indépendantes.  À l’exception des registres utilisés pour ren-

voyer une valeur de sortie, aucun registre   de donnée ou d’adresse   ne devra être modifié en sortie de  

vos sous-programmes. Attention     ! Tous les sous-programmes sont limités à   15   lignes d’instructions  

au maximum.

Structure d’un bitmap :

Champ
Taille

(bits)
Codage Description

WIDTH 16 Entier non signé Largeur du bitmap en pixels

HEIGHT 16 Entier non signé Hauteur du bitmap en pixels

MATRIX Variable Bitmap

Matrice de points du bitmap.

Si un bit est à 0, le pixel affiché est noir.

Si un bit est à 1, le pixel affiché est blanc.

Structure d’un sprite :

Champ
Taille

(bits)
Codage Description

STATE 16 Entier non signé
État d’affichage du sprite.

Seulement deux valeurs possibles : HIDE = 0 ou SHOW = 1

X 16 Entier signé Abscisse du sprite

Y 16 Entier signé Ordonnée du sprite

BITMAP1 32 Entier non signé Adresse du premier bitmap

BITMAP2 32 Entier non signé Adresse du second bitmap

On suppose que la taille du bitmap 1 est toujours égale a celle du bitmap 2.

Constantes déjà définies :

VIDEO_START         equ     $ffb500         ; Adresse de départ de la mémoire vidéo
VIDEO_SIZE          equ     (480*320/8)     ; Taille en octets de la mémoire vidéo

WIDTH               equ     0
HEIGHT              equ     2
MATRIX              equ     4

STATE               equ     0
X                   equ     2
Y                   equ     4
BITMAP1             equ     6
BITMAP2             equ     10

HIDE                equ     0
SHOW                equ     1

1. Réalisez le sous-programme FillScreen qui remplit la mémoire vidéo d’une valeur numérique. Le

remplissage se fera par mot de 32 bits.

Entrée : D0.L = Valeur numérique sur 32 bits avec laquelle sera remplie la mémoire vidéo.

2. Réalisez le sous-programme GetRectangle qui renvoie les coordonnées du rectangle qui délimite un

sprite.

Entrée : A0.L = Adresse d’un sprite.

Sorties : D1.W  = Abscisse du point supérieur gauche du sprite.

D2.W  = Ordonnée du point supérieur gauche du sprite.

D3.W  = Abscisse du point inférieur droit du sprite.

D4.W  = Ordonnée du point inférieur droit du sprite.

3. Réalisez le sous-programme MoveSprite qui déplace un sprite. Le déplacement se fera de façon rela-

tive. Si la nouvelle position du sprite fait sortir le sprite de l’écran, alors la position du sprite restera

inchangée (la nouvelle position ne sera pas prise en compte).

Entrées : A1.L = Adresse du sprite.

D1.W = Mouvement relatif horizontal en pixels (entier signé sur 16 bits).

D2.W = Mouvement relatif vertical en pixels (entier signé sur 16 bits).

Sorties : D0.L renvoie false (0) si le sprite n’a pas été déplacé (car cela le faisait sortir de l’écran).

D0.L renvoie true (1) si le sprite a été déplacé.

Pour  savoir  si  un  sprite  sort  de  l’écran,  vous  pouvez  effectuer  un  appel  au  sous-programme

IsOutOfScreen.  On  supposera  que  ce  sous-programme existe  déjà  (vous  n’avez  pas  besoin  de

l’écrire).

Entrées : A0.L = Adresse du bitmap.

D1.W = Abscisse en pixels du bitmap (entier signé sur 16 bits).

D2.W = Ordonnée en pixels du bitmap (entier signé sur 16 bits).

Sorties : Z renvoie false (0) si le bitmap ne sort pas de l’écran.

Z renvoie true (1) si le bitmap sort de l’écran.

Nom : .................................................... Prénom : .............................................. Classe : ........................... 

DOCUMENT RÉPONSE À RENDRE

Exercice     1  

Instruction Mémoire Registre

Exemple $005000  54 AF 00 40 E7 21 48 C0
A0 = $00005004

A1 = $0000500C

Exemple $005008  C9 10 11 C8 D4 36 FF 88 Aucun changement

MOVE.L #4507,-(A1) $005000  54 AF 18 B9 00 00 11 9B A1 = $00005004

MOVE.B $5009,-6(A1) $005000  54 AF 10 B9 E7 21 48 C0 Aucun changement

MOVE.W 8(A1),-37(A2,D0.W) $005000  13 79 18 B9 E7 21 48 C0 Aucun changement

MOVE.L -4(A2),$21(A0,D2.L) $005000  54 AF D4 36 1F 88 48 C0 Aucun changement

Exercice     2  

Opération
Taille

(bits)

Nombre manquant

(hexadécimal)
N Z V C

$80 + $? 8 $00 1 0 0 0

$8000 + $? 16 $8000 0 1 1 1

$80000000 + $? 32 $80000001 0 0 1 1

Exercice     3  

Valeurs des registres après exécution du programme.

Utilisez la représentation hexadécimale sur 32 bits.

D1 = $00000002 D3 = $00000800

D2 = $51005A80 D4 = $00001001

Exercice     4  

FillScreen          movem.l d7/a0,-(a7)

                    lea     VIDEO_START,a0
                    move.w  #VIDEO_SIZE/4-1,d7
                    
\loop               move.l  d0,(a0)+
                    dbra    d7,\loop
                    
                    movem.l (a7)+,d7/a0
                    rts

GetRectangle        move.l  a0,-(a7)

                    move.w  X(a0),d1
                    move.w  Y(a0),d2

                    movea.l BITMAP1(a0),a0

                    move.w  WIDTH(a0),d3
                    add.w   d1,d3
                    subq.w  #1,d3

                    move.w  HEIGHT(a0),d4
                    add.w   d2,d4
                    subq.w  #1,d4

                    movea.l (a7)+,a0
                    rts

MoveSprite          movem.l d1/d2/a0,-(a7)

                    add.w   X(a1),d1
                    add.w   Y(a1),d2

                    movea.l BITMAP1(a1),a0

                    jsr     IsOutOfScreen
                    beq     \false

                    move.w  d1,X(a1)
                    move.w  d2,Y(a1)

\true               moveq.l #1,d0
                    bra     \quit

\false              moveq.l #0,d0
\quit               movem.l (a7)+,d1/d2/a0
                    rts

